

Macros

Can I retrieve metadata from an empty dataset?

When a dataset contains values, the normal experience is that the dataset is processed sequentially, and values can, for example, be retrieved and stored in macro variables.

Consider the code:

```
data class ;
  set sashelp.class ;
run ;

data class2 ;
  set class (obs = 1) nobs = obcount ;
  call symputx('obcount',obcount) ;
  call symputx('age'      ,age)      ;
run ;

%put &=obcount
    &=age
    ;
```

Which returns values in the LOG for the OBCOUNT variable retrieved from metadata, and the AGE variable retrieved from data:

```
NOBS=19      AGE=14
```

However if the code is attempted against an empty dataset:

```
data class ;
  if 0 then set sashelp.class ;
stop ;
run ;

data class2 ;
  set class (obs = 1) nobs = obcount ;
  call symputx('obcount',obcount) ;
  call symputx('age'      ,age)      ;
run ;

%put &=obcount
    &=age
    ;
```

Macros

the LOG contains neither value, despite the stored metadata value of '0' for the record count being available.

WARNING: Apparent symbolic reference OBCOUNT not resolved.

WARNING: Apparent symbolic reference AGE not resolved.

It is the placement of the CALL routine below the SET statement which is causing the issue here. As there is no sequential processing of the dataset listed on the SET statement, the additional statements are not processed.

Because the OBS= stored value is available to the system at COMPILE time, the CALL routing retrieving the value can be placed above the SET statement, and avoid being ignored:

```
data class ;  
  if 0 then set sashelp.class ;  
stop ;  
run ;  
  
data class2 ;  
  call symputx('obcount',obcount) ;  
  set class (obs = 1) nobs = obcount ;  
  call symputx('age'      ,age)      ;  
run ;  
  
%put &=obcount  
    &=age  
    ;
```

Unique solution ID: #1052

Author: Alan D Rudland

Last update: 2020-01-09 16:27